


Robot Car

By **abdulsamad abdulsamad** - March 14, 2019



Download as PDF



Print

Today lets build a robot car

First we will decide what functions we want in our car. I want following functions;

1. **Manual Control**
2. **Line following**
3. **Obstacle avoidance**
4. **Colorful LEDs animation**
5. **Buzzer**
6. **Mobile control using Bluetooth Communication**
7. **Rechargeable battery**

Manual control

Lets start with simple design of car and controlling it. For the movement of car I will be using two dc gear motor. I will use a H Bridge to control the speed and direction of motor.

Truth table for this H-Bridge(L2983D) is as follow;

DC MOTOR	MODE	IN1	IN2	IN3	IN4
MOTOR-A	Forward	1/PWM	0		
	Reversion	0	1/PWM		
	Standby	0	0		
	Brake	1	1		
MOTOR-B	Forward			1/PWM	0
	Reversion			0	1/PWM
	Standby			0	0
	Brake			1	1




We will give this signal to H-Bridge using arduino pro mini. We will connect H-bridge to arduino pins 3,5,6 and 11 as they have pwm which we need for speed control.

In arduino IDE we will first define the pins and create a function to convert joystick values to H-bridge value according to its truth table.

Buzzer

For buzzer we will simply connect a mini buzzer to pin 7 of arduino.

It will look something like this.



Colorful RGB animation.

For RGB lights i will be using neopixel Ws281b rgb led strips. It only uses one data pin to control the entire strip. I am using pin 10 to control RGB strip. I am using adafruit neopixel library to control this strips.


Line following

For line following i will be using two IR proximity sensor. I have a tutorial on how these sensor works, check them out if dont how its working.

<https://stem.pk/what-is-infrared-sensor-ir-sensor/>



You can buy them commercially but i made my own. I will hookup them to pin 4 and 8. when one sensor is on black line it will turn in one way and when the other sensor is on black it will turn in the other way, otherwise it will go straight.



Rechargeable battery

For powering the car i will be using Li-ion battery of 2000mAH and 3.7v which will be boosted to 5v using MT3608 boost converter. This battery can last for quite a long time .


For charging I will use TP4056 charging module. This module charges the battery and as well as protects the battery from over-charging, over-discharging, over-current and short circuit protection. You can use any 5v mobile charger to charge battery.

I am also adding 10k resistor between battery + terminal and analog 0 pin of arduino to later get remaining battery.



There is a small problem, when you start the car from rest position and accelerate it quickly. A huge surge current passes through this module which consider it as over-current and cut of the power, to address this problem we have to see how this module is detecting over-current.

The charging module have following functional circuit;




By looking at these pictures we came to know that charging module have Dw01A protection chip, which senses voltage drop across power mosfets at CS pin. So we will add a 1k resistor at CS pin GND. This will make a voltage divider and will lower the sensing

voltage at higher currents.


You can also check this forum.

<https://electronics.stackexchange.com/questions/272947/remove-over-current-protection-of-battery-protection-circuit>

Functional Block Diagram



Typical Application Circuit



Obstacle avoiding

To detect obstacle we will need a ultrasonic sensor which uses sonar to detect objects. We will mount it on servo motor so that we get 180° of view. In this mode normally car will move straight until it detects some object. when it detects some object, it will stop look left and right and will turn to the position it see no object.



Another problem occurs when we use servo. The problem is servo does not work with neopixel. Why that's is happening? for that we have to see how these libraries work. Both libraries uses 16 bit timer of the arduino to make time interrupts which are required for normal operation of servo and neo pixel and unfortunately arduino pro mini which uses Atmega328p only have one 16 bit timer *i.e* timer 1 that means both libraries conflict with each other. To solve this problem we have to use TicoServo library for servo which uses pwm mode of timer 1 to generate signal required for servo.


If you want details on this topic you can check this link.

<https://learn.adafruit.com/neopixels-and-servos/overview>


Mobile control

For mobile control we want an app to communicate to arduino using bluetooth. I will be using RemoteXY for that. You can find it on Play Store.

GUI of app will be made on RemoteXY [website](#). We will add joystick to control car, bar to show remaining battery, slide switch to select current driving mode, a button for buzzer and finally a color wheel for RGB lights. Of course you can add according to your taste



and our beautiful car is ready.



Complete code is here;

```

1.  #include <Ultrasonic.h>
2.  #include <Adafruit_TiCoServo.h>
3.  #include <Adafruit_NeoPixel.h>
4.  #ifdef __AVR__
5.      #include <avr/power.h>
6.  #endif
7.
8.  // Which pin on the Arduino is connected to the NeoPixels?
9.  // On a Trinket or Gemma we suggest changing this to 1
10. #define PIN          10
11. int irsensor1 = 4;
12. int irsensor2 = 8;
13. //define right motor control pins
14. #define int1 3
15. #define int2 5
16.
17. //define left motor control pins
18. #define int3 6
19. #define int4 11
20.
21. //define two arrays with a list of pins for each motor
22. uint8_t RightMotor[2] = {int1, int2};
23. uint8_t LeftMotor[2] = {int3, int4};
24.
25. // How many NeoPixels are attached to the Arduino?
26. #define NUMPIXELS     11
27.
28. // When we setup the NeoPixel library, we tell it how many pixels, and which pin to use to send
29. // signals.
30. // Note that for older NeoPixel strips you might need to change the third parameter--see the
31. // strandtest
32. // example for more information on possible values.
33. Adafruit_NeoPixel pixels = Adafruit_NeoPixel(NUMPIXELS, PIN, NEO_GRB + NEO_KHZ800);
34.
35. ///////////////////////////////////////////////////////////////////
36. // RemoteXY include library //
37. ///////////////////////////////////////////////////////////////////
38.
39. // RemoteXY select connection mode and include library
40. #define REMOTEXY_MODE __HARDSERIAL
41.
42. #include <RemoteXY.h>
43.
44. // RemoteXY connection settings
45. #define REMOTEXY_SERIAL Serial
46. #define REMOTEXY_SERIAL_SPEED 9600
47.
48. // RemoteXY configurate
49. #pragma pack(push, 1)
50. uint8_t RemoteXY_CONF[] =
51. { 255,7,0,1,0,46,0,8,24,0,
52. 5,52,45,9,43,43,31,209,24,3,
53. 3,88,35,10,26,94,14,6,0,3,
54. 33,25,25,24,14,1,0,36,41,12,
55. 12,35,33,72,0,66,0,4,4,9,
56. 21,77,24 };
57.
58. // this structure defines all the variables of your control interface
59. struct {
60.
61.     // input variable
62.     int8_t joystick_x; // =-100..100 x-coordinate joystick position
63.     int8_t joystick_y; // =-100..100 y-coordinate joystick position
64.     uint8_t slider; // =0 if select position A, =1 if position B, =2 if position C, ...
65.     uint8_t rgb_r; // =0..255 Red color value
66.     uint8_t rgb_g; // =0..255 Green color value
67.     uint8_t rgb_b; // =0..255 Blue color value
68.     uint8_t button; // =1 if button pressed, else =0
69.
70.     // output variable
71.     int8_t battery; // =0..100 level position
72.
73.     // other variable
74.     uint8_t connect_flag; // =1 if wire connected, else =0
75.
76. } RemoteXY;
77. #pragma pack(pop)

```

```

78.  //////////////////////////////////////
79.  //          END RemoteXY include          //
80.  //////////////////////////////////////
81.
82.  #define SERVO_PIN    9
83.  #define SERVO_MIN 550 // 1 ms pulse
84.  #define SERVO_MAX 2150 // 2 ms pulse
85.  Adafruit_TiCoServo servo;
86.
87.  int n=0;
88.  long unsigned battery_volts=0;
89.  int battery_pin = A3;
90.  #define PIN_BUTTON 7
91.
92.  Ultrasonic ultrasonic(12, 13);
93.  int distancel;
94.  int distancer;
95.  int distancef;
96.  int stuck = false;
97.
98.  void setup()
99.  {
100.     RemoteXY_Init ();
101.     pixels.begin();
102.     servo.attach(SERVO_PIN, SERVO_MIN, SERVO_MAX);
103.     pinMode (PIN_BUTTON, OUTPUT);
104.     pinMode (int1, OUTPUT);
105.     pinMode (int2, OUTPUT);
106.     pinMode (int3, OUTPUT);
107.     pinMode (int4, OUTPUT);
108.     pinMode (irsensor1, INPUT);
109.     pinMode (irsensor2, INPUT);
110.     RemoteXY.battery = map(analogRead(battery_pin), 500 , 890 ,0 , 100); //read battery voltage
111. }
112.
113. void loop()
114. {
115.     RemoteXY_Handler (); // use the RemoteXY structure for data transfer
116.     digitalWrite(PIN_BUTTON, (RemoteXY.button==0)?LOW:HIGH); //turn buzzer on or off
117.     servo.write(1350); //set servo postion at centre
118.     //taking multiple readings for average readings of battery voltage
119.     battery_volts = analogRead(battery_pin) + battery_volts;
120.     n++;
121.     if(n==1000){
122.         battery_volts = battery_volts/1000;
123.         n=0;
124.         RemoteXY.battery = map(battery_volts, 500 , 890 ,0 , 100);
125.         battery_volts=0;
126.     }
127.     //setting rgb color
128.     for(int i=0;i<NUMPIXELS;i++){
129.         // pixels.Color takes RGB values, from 0,0,0 up to 255,255,255
130.         pixels.setPixelColor(i, pixels.Color(RemoteXY.rgb_r,RemoteXY.rgb_g,RemoteXY.rgb_b));
131.         pixels.show(); // This sends the updated pixel color to the hardware.
132.     }
133.     //selecting modes for driving
134.     if(RemoteXY.slider == 0){
135.         if(RemoteXY.joystick_y > 0){
136.             jsteering (RightMotor, RemoteXY.joystick_y + RemoteXY.joystick_x);
137.             jsteering (LeftMotor, RemoteXY.joystick_y - RemoteXY.joystick_x);
138.         }
139.         else if(RemoteXY.joystick_y < 0){
140.             jsteering (RightMotor, RemoteXY.joystick_y - RemoteXY.joystick_x);
141.             jsteering (LeftMotor, RemoteXY.joystick_y + RemoteXY.joystick_x);
142.         }
143.         else{
144.             digitalWrite (RightMotor [0], LOW);
145.             digitalWrite (RightMotor [1], LOW);
146.             digitalWrite (LeftMotor [0], LOW);
147.             digitalWrite (LeftMotor [1], LOW);
148.         }
149.     }
150.     else if(RemoteXY.slider == 1){
151.         irsteering();
152.     }
153.     else if(RemoteXY.slider == 2){
154.         ussteering();
155.     }
156. }

```

```

157.
158. void jsteering (uint8_t * motor, int v) // v = motor speed, motor = pointer to an array of pins
159. {
160.     if (v > 100) v=100;
161.     if (v < -100) v=-100;
162.     if (v > 0){
163.         analogWrite (motor [0], v * 2.55);
164.         digitalWrite (motor [1], LOW);
165.     }
166.     else if ( v<0 ){
167.         digitalWrite (motor [0], LOW);
168.         analogWrite (motor [1], (-v) * 2.55);
169.     }
170.     else{
171.         digitalWrite (motor [0], LOW);
172.         digitalWrite (motor [1], LOW);
173.         analogWrite (motor [2], 0);
174.     }
175. }
176.
177. void irsteering(){
178.     if(digitalRead(irsensor1)==HIGH && digitalRead(irsensor2)==HIGH)
179.     {
180.
181.         digitalWrite(int2, 0);
182.         digitalWrite(int4, 0);
183.         analogWrite(int1,100);
184.         analogWrite(int3,100);
185.     }
186.     else if(digitalRead(irsensor1)==HIGH)
187.     {
188.         digitalWrite(int2, 0);
189.         digitalWrite(int4, 0);
190.         analogWrite(int1,50);
191.         analogWrite(int3,100);
192.     }
193.     else if(digitalRead(irsensor2)==HIGH)
194.     {
195.         digitalWrite(int2, 0);
196.         digitalWrite(int4, 0);
197.         analogWrite(int1,100);
198.         analogWrite(int3,50);
199.     }
200.     else{
201.         digitalWrite(int2, 0);
202.         digitalWrite(int4, 0);
203.         digitalWrite(int1,0);
204.         digitalWrite(int3,0);
205.     }
206. }
207.
208. void ussteering(){
209.     servo.write(1350);
210.     if(stuck == true){
211.         delay(500);
212.         stuck = false;
213.     }
214.     distancef = ultrasonic.read();
215.     if (distancef < 20){
216.         stuck = true;
217.         analogWrite(int1, 0);
218.         analogWrite(int2, 0);
219.         analogWrite(int3, 0);
220.         analogWrite(int4, 0);
221.         delay(500);
222.         servo.write(550);
223.         delay(500);
224.         distancel = ultrasonic.read();
225.         servo.write(2150);
226.         delay(500);
227.         distancer = ultrasonic.read();
228.         if(distancel > distancer && distancel > 20){
229.             digitalWrite(int1, LOW);
230.             digitalWrite(int4, LOW);
231.             analogWrite(int3,200);
232.             analogWrite(int2,200);
233.             delay(50);
234.         }
235.         else if(distancer > distancel && distancer > 20){

```




```

236.         digitalWrite(int2, LOW);
237.         digitalWrite(int3, LOW);
238.         analogWrite(int1,200);
239.         analogWrite(int4,200);
240.         delay(50);
241.     }
242.     else{
243.         digitalWrite(int2, LOW);
244.         digitalWrite(int3, LOW);
245.         analogWrite(int1,200);
246.         analogWrite(int4,200);
247.         delay(100);
248.     }
249. }
250. else if (distancef > 20){
251.     analogWrite(int2, 0);
252.     analogWrite(int4, 0);
253.     analogWrite(int1,200);
254.     analogWrite(int3,200);
255. }
256. }

```

schematic looks something like this



Here is where i got my parts from

2xDC gear motors <https://hallroad.org/smart-robot-car-plastic-tire-tyre-wheel-dc-3-6v-gear-motor-set.html>

2xIR sensor <https://hallroad.org/ir-infrared-obstacle-avoidance-sensor-module-in-pakistan.html>

Buzzer <https://hallroad.org/12v-5v-piezo-buzzer-active-buzzer-continous-beep.html>

Servo SG90 <https://hallroad.org/towerpro-sg90-sg-90-180-degree-degree-servo-motor-in-pakistan.html>

Neopixel RGB strip <https://hallroad.org/dc5v-ws2812b-5-meter-60-led-strip-addressable-leds-in-pakistan-en.html>

Ultrasonic sensor <https://hallroad.org/hc-sr04-hc-sr04-ultrasonic-sensor-ultrasonic-distance-sensor-in-pakistan.html>

Ultrasonic sensor bracket <https://hallroad.org/ultrasonic-distance-sensor-mounting-bracket-in-pakistan.html>

Arduino pro mini <https://hallroad.org/arduino-pro-mini-in-pakistan.html>

Mini H-Bridge <https://hallroad.org/1.5a-mini-dual-channel-dc-motor-driver-module-l298n-pwm-speed-controller-in-pakistan.html>

Battery 2000mAh 3.7v <https://hallroad.org/2000mah-3.7v-lithium-ion-battery-li-ion-battery-with-jst2.0-connector-in-pakistan.html>

MT3608 Boost converter <https://hallroad.org/mt3608-2a-max-dc-dc-step-up-power-module-booster-power-module-in-pakistan.html>

TP4056 charging and protection module <https://hallroad.org/3.7v-li-ion-charger-module-1a-with-battery-protection-tp4056-in-pakistan.html>

Bluetooth HC-05 <https://hallroad.org/hc-05-bluetooth-module-in-pakistan.html>

Toggle switch <https://hallroad.org/toggle-switch-spdt-on-off-in-pakistan.html>

Caster Wheel <https://hallroad.org/metal-ball-caster-wheel-20mm-metal-ball-for-robot.html>

Robot Car chassis <https://hallroad.org/2wd-round-robot-car-chassis-in-pakistan.html>

abdulsamad abdulsamad